
Abstract

This development formalizes the Busy Beaver upper-bound princi-
ple and its computability consequences for the Turing machines from
AFP’s Universal Turing Machine entry. The abstract layer works over
machine models with finite size classes and unique exact halting times:
it defines the time Busy Beaver function and proves that any total
upper bound for this function decides zero-input halting. The devel-
opment then connects the abstract theory to AFP’s Universal Tur-
ing Machine formalization by defining exact numeric-result halting
times, a finite size measure for Turing programs, a concrete Turing-
machine Busy Beaver function, a code-indexed Busy Beaver function
whose upper bounds decide AFP’s special halting problem K1, and a
strengthened program/input-pair Busy Beaver function whose upper
bounds decide AFP’s two-argument halting problem H1. Using AFP’s
Turing-computability interface, it further proves that no upper-bound-
induced H1 decider set has a Turing-computable total characteristic
function, with a specialization to the decider induced by the concrete
pair-indexed Busy Beaver function itself.

3 The Busy Beaver Entry 10

Main Results
The formalization separates a reusable Busy Beaver upper-bound principle
from its concrete Turing-machine applications.

• Any total upper bound for the abstract function BB-time decides zero-
input halting for the corresponding model.

• The concrete function Turing-BB-time bounds exact numeric-result
halting times for bounded-size Turing programs.

• In AFP’s hpk locale, code-indexed upper bounds decide the diagonal
halting problem K1.

• Pair-indexed upper bounds decide the two-argument halting prob-
lem H1, and the induced H1 characteristic functions are not Turing-
computable total.

theory Busy-Beaver-Base
imports Main

begin

1 The Busy Beaver Upper-Bound Principle
Radó’s Busy Beaver function measures the largest halting time among ma-
chines of bounded size [1]. This theory formalizes the Busy Beaver upper-
bound principle over a model-parametric notion of programs with a finite
size measure and unique exact halting times.

The base locale defines the Busy Beaver time function for zero-input
runs and proves that any total upper bound for it decides the corresponding
fixed-input halting predicate. A second locale adds a generic computability
interface: noncomputability follows from an assumed absence of computable
fixed-input halting deciders, and eventual domination is derived under an
explicit witness assumption that packages the usual input-to-program com-
pilation argument. A separate theory instantiates the base locale with AFP’s
Turing machines and, by treating program/input pairs as Busy Beaver ob-
jects, obtains a concrete upper-bound decider for AFP’s two-argument halt-
ing problem together with an explicit Turing-noncomputability consequence
for the induced characteristic functions.

1.1 Base model and Busy Beaver time
locale busy-beaver-base =

fixes size :: ′prog ⇒ nat
and halts-in :: ′prog ⇒ nat ⇒ nat ⇒ bool

assumes finite-size-le: finite {p. size p ≤ n}
and halting-time-unique:

halts-in p x t =⇒ halts-in p x u =⇒ t = u
begin

definition halts :: ′prog ⇒ nat ⇒ bool where
halts p x ←→ (∃ t. halts-in p x t)

definition halting-time :: ′prog ⇒ nat ⇒ nat where
halting-time p x = (THE t. halts-in p x t)

definition time-set :: nat ⇒ nat set where
time-set n = {t. ∃ p. size p ≤ n ∧ halts-in p 0 t}

definition BB-time :: nat ⇒ nat where
BB-time n = Max (insert 0 (time-set n))

definition upper-bound-for-BB :: (nat ⇒ nat) ⇒ bool where
upper-bound-for-BB b ←→ (∀n. BB-time n ≤ b n)

definition halting-decider-from :: (nat ⇒ nat) ⇒ ′prog ⇒ nat ⇒ bool where
halting-decider-from b p x ←→ (∃ t≤b (size p). halts-in p x t)

lemma halting-time-eq:
assumes halts-in p x t
shows halting-time p x = t
〈proof 〉

lemma finite-time-set [simp]: finite (time-set n)
〈proof 〉

lemma finite-insert-time-set [simp]: finite (insert 0 (time-set n))
〈proof 〉

lemma BB-time-ge-time:
assumes size p ≤ n

and halts-in p 0 t
shows t ≤ BB-time n
〈proof 〉

lemma BB-time-upper-bound:
assumes

∧
p t. size p ≤ n =⇒ halts-in p 0 t =⇒ t ≤ B

shows BB-time n ≤ B
〈proof 〉

lemma halting-decider-from-sound:
assumes halting-decider-from b p x
shows halts p x
〈proof 〉

lemma halting-decider-from-complete-0 :
assumes ub: upper-bound-for-BB b
assumes halts p 0
shows halting-decider-from b p 0
〈proof 〉

theorem halting-decider-from-correct-0 :
assumes upper-bound-for-BB b
shows halting-decider-from b p 0 ←→ halts p 0
〈proof 〉

lemma BB-time-is-upper-bound: upper-bound-for-BB BB-time
〈proof 〉

end

1.2 Computability consequences
The following locale deliberately keeps computability assumptions explicit.
In particular, eventual domination is not derived from finite size classes
and exact halting times alone; the assumption computable-has-busy-witness
states the compilation witness needed to turn values of a computed function
into sufficiently long zero-input halting computations.
locale busy-beaver-model = busy-beaver-base size halts-in

for size :: ′prog ⇒ nat
and halts-in :: ′prog ⇒ nat ⇒ nat ⇒ bool +

fixes computes :: ′prog ⇒ (nat ⇒ nat) ⇒ bool
assumes compute-halts:

computes p f =⇒ ∃ t. halts-in p x t
and computable-has-busy-witness:

computes p f =⇒ ∃N . ∀n≥N .
∃ q t. size q ≤ n ∧ halts-in q 0 t ∧ f n ≤ t

begin

definition computable :: (nat ⇒ nat) ⇒ bool where
computable f ←→ (∃ p. computes p f )

definition eventually-dominates :: (nat ⇒ nat) ⇒ (nat ⇒ nat) ⇒ bool where
eventually-dominates g f ←→ (∃N . ∀n≥N . f n ≤ g n)

lemma no-computable-upper-bound-if-halting-undecidable:
assumes no-decider :∧

b. computable b =⇒

¬ (∀ p. halting-decider-from b p 0 ←→ halts p 0 )
shows ¬ (∃ b. computable b ∧ upper-bound-for-BB b)
〈proof 〉

theorem BB-time-not-computable-if-halting-undecidable:
assumes no-decider :∧

b. computable b =⇒
¬ (∀ p. halting-decider-from b p 0 ←→ halts p 0 )

shows ¬ computable BB-time
〈proof 〉

lemma BB-time-dominates-computed-function-from-size:
assumes comp: computes p f
shows eventually-dominates BB-time f
〈proof 〉

theorem BB-time-eventually-dominates-computable:
assumes computable f
shows eventually-dominates BB-time f
〈proof 〉

end

end
theory Turing-Busy-Beaver

imports
Busy-Beaver-Base
Universal-Turing-Machine.TuringComputable

begin

2 Busy Beaver Upper Bounds for Universal Tur-
ing Machines

This theory connects the abstract Busy Beaver upper-bound principle to the
Turing machines from AFP’s Universal Turing Machine entry [2]. We use
halting with a numeric result, the halting predicate that occurs in AFP’s
formalization of the special halting problems. The first instantiation is a
direct Turing-program Busy Beaver function. Inside AFP’s hpk locale we
also define code-indexed variants for the one-argument problem K1 and
the two-argument problem H1. Finally, using AFP’s Turing-computability
interface, we make the noncomputability consequence explicit for the char-
acteristic functions of the induced H1 decider sets.

Three related Busy Beaver notions appear below. The theory Busy-Beaver-Base
supplies the model-parametric function BB-time, whose objects are abstract
programs run on input 0. The first concrete interpretation instantiates this
as Turing-BB-time for AFP Turing programs, still using a direct program-

size measure. The code-indexed interpretations are introduced for the halt-
ing problems in the Universal Turing Machine entry: the diagonal code-
indexed function decides K1, while the pair-indexed function folds a ma-
chine code and input into a single Busy Beaver object so that upper bounds
decide H1.

2.1 Exact halting times for numeric results
definition has-num-result-at :: tprog0 ⇒ nat list ⇒ nat ⇒ bool where

has-num-result-at tm ns t ←→
is-final (steps0 (1 , [], <ns>) tm t) ∧
(∃ k n l. steps0 (1 , [], <ns>) tm t = (0 , Bk ↑ k, <n::nat> @ Bk ↑ l))

definition tm-num-halts-in :: tprog0 ⇒ nat ⇒ nat ⇒ bool where
tm-num-halts-in tm input t ←→

has-num-result-at tm [input] t ∧ (∀ u<t. ¬ has-num-result-at tm [input] u)

lemma tm-num-halts-in-unique:
assumes tm-num-halts-in tm input t

and tm-num-halts-in tm input u
shows t = u
〈proof 〉

lemma TMC-has-num-res-iff-tm-num-halts:
TMC-has-num-res tm [input] ←→ (∃ t. tm-num-halts-in tm input t)
〈proof 〉

2.2 A finite size measure for Turing programs
fun action-rank :: action ⇒ nat where

action-rank WB = 0
| action-rank WO = 1
| action-rank L = 2
| action-rank R = 3
| action-rank Nop = 4

definition instr-size :: instr ⇒ nat where
instr-size i = Suc (action-rank (fst i) + snd i)

definition tm-size :: tprog0 ⇒ nat where
tm-size tm = length tm + sum-list (map instr-size tm)

lemma finite-instr-size-le: finite {i::instr . instr-size i ≤ n}
〈proof 〉

lemma length-le-tm-size:
length tm ≤ tm-size tm
〈proof 〉

lemma instr-size-le-tm-size:
assumes i ∈ set tm
shows instr-size i ≤ tm-size tm
〈proof 〉

lemma finite-tm-size-le: finite {tm::tprog0 . tm-size tm ≤ n}
〈proof 〉

interpretation turing-busy-beaver : busy-beaver-base tm-size tm-num-halts-in
〈proof 〉

abbreviation Turing-BB-time :: nat ⇒ nat where
Turing-BB-time ≡ turing-busy-beaver .BB-time

theorem Turing-BB-time-ge:
assumes tm-size tm ≤ n

and tm-num-halts-in tm 0 t
shows t ≤ Turing-BB-time n
〈proof 〉

2.3 The code-indexed Busy Beaver function and AFP’s spe-
cial halting problem

context hpk
begin

The next predicate is deliberately diagonal: the formal locale input is
not used because the special halting problem K1 asks whether the machine
coded by c halts on input c.
definition coded-diagonal-num-halts-in :: nat ⇒ nat ⇒ nat ⇒ bool where

coded-diagonal-num-halts-in c input t ←→ tm-num-halts-in (c2t c) c t

interpretation coded-busy-beaver : busy-beaver-base λc::nat. c coded-diagonal-num-halts-in
〈proof 〉

abbreviation Coded-BB-time :: nat ⇒ nat where
Coded-BB-time ≡ coded-busy-beaver .BB-time

lemma coded-halts-iff-K1 :
coded-busy-beaver .halts c 0 ←→ [c] ∈ K1
〈proof 〉

theorem coded-BB-upper-bound-decides-K1 :
assumes coded-busy-beaver .upper-bound-for-BB b
shows coded-busy-beaver .halting-decider-from b c 0 ←→ [c] ∈ K1
〈proof 〉

corollary K1-not-turing-decidable-again: ¬ turing-decidable K1
〈proof 〉

2.4 A code-indexed Busy Beaver function for the two-argument
halting problem

The previous code-indexed function targets the diagonal/special problem
K1. To cover AFP’s two-argument halting problem H1, we fold the input
into the Busy Beaver object itself: an object is a pair (c, m) consisting of
a machine code and an input. The size measure c + m has finite bounded
classes, and an upper bound for the resulting Busy Beaver function decides
membership in H1.
definition coded-pair-num-halts-in :: (nat × nat) ⇒ nat ⇒ nat ⇒ bool where

coded-pair-num-halts-in cm input t ←→
tm-num-halts-in (c2t (fst cm)) (snd cm) t

definition coded-pair-size :: nat × nat ⇒ nat where
coded-pair-size cm = fst cm + snd cm

lemma finite-coded-pair-size-le:
finite {cm::nat × nat. coded-pair-size cm ≤ n}
〈proof 〉

interpretation pair-busy-beaver : busy-beaver-base coded-pair-size coded-pair-num-halts-in
〈proof 〉

abbreviation Pair-BB-time :: nat ⇒ nat where
Pair-BB-time ≡ pair-busy-beaver .BB-time

lemma coded-pair-halts-iff-H1 :
pair-busy-beaver .halts (c, m) 0 ←→ [c, m] ∈ H1
〈proof 〉

theorem pair-BB-upper-bound-decides-H1-pair :
assumes pair-busy-beaver .upper-bound-for-BB b
shows pair-busy-beaver .halting-decider-from b (c, m) 0 ←→ [c, m] ∈ H1
〈proof 〉

definition H1-decider-from :: (nat ⇒ nat) ⇒ nat list ⇒ bool where
H1-decider-from b nl ←→
(case nl of
[c, m] ⇒ pair-busy-beaver .halting-decider-from b (c, m) 0
| - ⇒ False)

theorem H1-decider-from-correct:
assumes pair-busy-beaver .upper-bound-for-BB b
shows H1-decider-from b nl ←→ nl ∈ H1
〈proof 〉

corollary H1-not-turing-decidable-again: ¬ turing-decidable H1
〈proof 〉

For every upper bound, the Boolean decider above defines the same set
as H1. Hence AFP’s existing undecidability theorem for H1 immediately
yields a concrete noncomputability consequence: the characteristic function
of any such upper-bound-induced decider set is not Turing-computable, even
in the total sense. In particular, this applies to the decider induced by
Pair-BB-time itself.
definition H1-decider-set-from :: (nat ⇒ nat) ⇒ nat list set where

H1-decider-set-from b = {nl. H1-decider-from b nl}

lemma H1-decider-set-from-eq-H1 :
assumes pair-busy-beaver .upper-bound-for-BB b
shows H1-decider-set-from b = H1
〈proof 〉

theorem no-turing-decidable-H1-decider-set-from:
assumes pair-busy-beaver .upper-bound-for-BB b
shows ¬ turing-decidable (H1-decider-set-from b)
〈proof 〉

theorem no-turing-computable-partial-H1-decider-from:
assumes pair-busy-beaver .upper-bound-for-BB b
shows ¬ turing-computable-partial (chi-fun (H1-decider-set-from b))
〈proof 〉

theorem no-turing-computable-total-H1-decider-from:
assumes pair-busy-beaver .upper-bound-for-BB b
shows ¬ turing-computable-total (chi-fun (H1-decider-set-from b))
〈proof 〉

theorem no-turing-computable-total-H1-deciding-upper-bound:
¬ (∃ b. pair-busy-beaver .upper-bound-for-BB b ∧

turing-computable-total (chi-fun (H1-decider-set-from b)))
〈proof 〉

corollary no-turing-computable-total-Pair-BB-time-decider :
¬ turing-computable-total (chi-fun (H1-decider-set-from Pair-BB-time))
〈proof 〉

end

end
theory Busy-Beaver

imports
Busy-Beaver-Base
Turing-Busy-Beaver

begin

3 The Busy Beaver Entry
This theory is the AFP entry point. It collects the abstract Busy Beaver
upper-bound principle and its instantiation for the Turing machines from
the AFP Universal Turing Machine development.
end

AI Assistance
AI assistance was used for proof engineering. The final definitions, state-
ments, and proofs are checked by Isabelle.

